NAG Library Routine Document

F08QYF (ZTRSNA)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08QYF (ZTRSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of a complex upper triangular matrix.

2 Specification

```
SUBROUTINE F08QYF (JOB, HOWMNY, SELECT, N, T, LDT, VL, LDVL, VR, LDVR, S,       &
                   SEP, MM, M, WORK, LDWORK, RWORK, INFO)

INTEGER                 N, LDT, LDVL, LDVR, MM, M, LDWORK, INFO
REAL (KIND=nag_wp)      S(*), SEP(*), RWORK(*)
COMPLEX (KIND=nag_wp) T(LDT,*), VL(LDVL,*), VR(LDVR,*), WORK(LDWORK,*)
LOGICAL                 SELECT(*)
CHARACTER(1)            JOB, HOWMNY
```

The routine may be called by its LAPACK name ***ztrsna***.

3 Description

F08QYF (ZTRSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of a complex upper triangular matrix T. These are the same as the condition numbers of the eigenvalues and right eigenvectors of an original matrix $A = ZTZ^{\mathrm{H}}$ (with unitary Z), from which T may have been derived.

F08QYF (ZTRSNA) computes the reciprocal of the condition number of an eigenvalue λ_i as

$$s_i = \frac{\left|v^{\mathrm{H}}u\right|}{\|u\|_E\|v\|_E},$$

where u and v are the right and left eigenvectors of T, respectively, corresponding to λ_i. This reciprocal condition number always lies between zero (i.e., ill-conditioned) and one (i.e., well-conditioned).

An approximate error estimate for a computed eigenvalue λ_i is then given by

$$\frac{\epsilon\|T\|}{s_i},$$

where ϵ is the ***machine precision***.

To estimate the reciprocal of the condition number of the right eigenvector corresponding to λ_i, the routine first calls F08QTF (ZTREXC) to reorder the eigenvalues so that λ_i is in the leading position:

$$T = Q\begin{pmatrix} \lambda_i & c^{\mathrm{H}} \\ 0 & T_{22} \end{pmatrix}Q^{\mathrm{H}}.$$

The reciprocal condition number of the eigenvector is then estimated as sep_i, the smallest singular value of the matrix $(T_{22} - \lambda_i I)$. This number ranges from zero (i.e., ill-conditioned) to very large (i.e., well-conditioned).

An approximate error estimate for a computed right eigenvector u corresponding to λ_i is then given by

$$\frac{\epsilon\|T\|}{sep_i}.$$

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOB – CHARACTER(1) *Input*

On entry: indicates whether condition numbers are required for eigenvalues and/or eigenvectors.

JOB = 'E'
Condition numbers for eigenvalues only are computed.

JOB = 'V'
Condition numbers for eigenvectors only are computed.

JOB = 'B'
Condition numbers for both eigenvalues and eigenvectors are computed.

Constraint: JOB = 'E', 'V' or 'B'.

2: HOWMNY – CHARACTER(1) *Input*

On entry: indicates how many condition numbers are to be computed.

HOWMNY = 'A'
Condition numbers for all eigenpairs are computed.

HOWMNY = 'S'
Condition numbers for selected eigenpairs (as specified by SELECT) are computed.

Constraint: HOWMNY = 'A' or 'S'.

3: SELECT($*$) – LOGICAL array *Input*

Note: the dimension of the array SELECT must be at least $\max(1, \mathrm{N})$ if HOWMNY = 'S', and at least 1 otherwise.

On entry: specifies the eigenpairs for which condition numbers are to be computed if HOWMNY = 'S'. To select condition numbers for the eigenpair corresponding to the eigenvalue λ_j, SELECT(j) must be set to .TRUE..

If HOWMNY = 'A', SELECT is not referenced.

4: N – INTEGER *Input*

On entry: n, the order of the matrix T.

Constraint: N ≥ 0.

5: T(LDT,$*$) – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array T must be at least $\max(1, \mathrm{N})$.

On entry: the n by n upper triangular matrix T, as returned by F08PSF (ZHSEQR).

6: LDT – INTEGER *Input*

On entry: the first dimension of the array T as declared in the (sub)program from which F08QYF (ZTRSNA) is called.

Constraint: LDT $\geq \max(1, \mathrm{N})$.

7: VL(LDVL,$*$) – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array VL must be at least $\max(1, \mathrm{MM})$ if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.

On entry: if JOB = 'E' or 'B', VL must contain the left eigenvectors of T (or of any matrix QTQ^{H} with Q unitary) corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors **must** be stored in consecutive columns of VL, as returned by F08PXF (ZHSEIN) or F08QXF (ZTREVC).

If JOB = 'V', VL is not referenced.

8: LDVL – INTEGER *Input*

On entry: the first dimension of the array VL as declared in the (sub)program from which F08QYF (ZTRSNA) is called.

Constraints:

if JOB = 'E' or 'B', LDVL $\geq \max(1, \mathrm{N})$;
if JOB = 'V', LDVL ≥ 1.

9: VR(LDVR,$*$) – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array VR must be at least $\max(1, \mathrm{MM})$ if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.

On entry: if JOB = 'E' or 'B', VR must contain the right eigenvectors of T (or of any matrix QTQ^{H} with Q unitary) corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors **must** be stored in consecutive columns of VR, as returned by F08PXF (ZHSEIN) or F08QXF (ZTREVC).

If JOB = 'V', VR is not referenced.

10: LDVR – INTEGER *Input*

On entry: the first dimension of the array VR as declared in the (sub)program from which F08QYF (ZTRSNA) is called.

Constraints:

if JOB = 'E' or 'B', LDVR $\geq \max(1, \mathrm{N})$;
if JOB = 'V', LDVR ≥ 1.

11: S($*$) – REAL (KIND=nag_wp) array *Output*

Note: the dimension of the array S must be at least $\max(1, \mathrm{MM})$ if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.

On exit: the reciprocal condition numbers of the selected eigenvalues if JOB = 'E' or 'B', stored in consecutive elements of the array. Thus S(j), SEP(j) and the jth columns of VL and VR all correspond to the same eigenpair (but not in general the jth eigenpair unless all eigenpairs have been selected).

S is not referenced if JOB = 'V'.

12: SEP($*$) – REAL (KIND=nag_wp) array *Output*

Note: the dimension of the array SEP must be at least $\max(1, \mathrm{MM})$ if JOB = 'V' or 'B' and at least 1 if JOB = 'E'.

On exit: the estimated reciprocal condition numbers of the selected right eigenvectors if JOB = 'V' or 'B', stored in consecutive elements of the array.

If JOB = 'E', SEP is not referenced i.

13: MM – INTEGER *Input*

On entry: the number of elements in the arrays S and SEP, and the number of columns in the arrays VL and VR (if used). The precise number required, m, is n if HOWMNY = 'A'; if HOWMNY = 'S', m is the number of selected eigenpairs (see SELECT), in which case $0 \leq m \leq n$.

Constraint: MM $\geq m$.

14: M – INTEGER *Output*

On exit: m, the number of selected eigenpairs. If HOWMNY = 'A', M is set to n.

15: WORK(LDWORK,$*$) – COMPLEX (KIND=nag_wp) array *Workspace*

Note: the second dimension of the array WORK must be at least $\max(1, \mathrm{N}+1)$ if JOB = 'V' or 'B' and at least 1 if JOB = 'E'.

If JOB = 'E', WORK is not referenced.

16: LDWORK – INTEGER *Input*

On entry: the first dimension of the array WORK as declared in the (sub)program from which F08QYF (ZTRSNA) is called.

Constraints:

if JOB = 'V' or 'B', LDWORK $\geq \max(1, \mathrm{N})$;
if JOB = 'E', LDWORK ≥ 1.

17: RWORK($*$) – REAL (KIND=nag_wp) array *Workspace*

Note: the dimension of the array RWORK must be at least $\max(1, \mathrm{N})$.

18: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed values sep_i may over estimate the true value, but seldom by a factor of more than 3.

8 Further Comments

The real analogue of this routine is F08QLF (DTRSNA).

9 Example

This example computes approximate error estimates for all the eigenvalues and right eigenvectors of the matrix T, where

$$T = \begin{pmatrix} -6.0004 - 6.9999i & 0.3637 - 0.3656i & -0.1880 + 0.4787i & 0.8785 - 0.2539i \\ 0.0000 + 0.0000i & -5.0000 + 2.0060i & -0.0307 - 0.7217i & -0.2290 + 0.1313i \\ 0.0000 + 0.0000i & 0.0000 + 0.0000i & 7.9982 - 0.9964i & 0.9357 + 0.5359i \\ 0.0000 + 0.0000i & 0.0000 + 0.0000i & 0.0000 + 0.0000i & 3.0023 - 3.9998i \end{pmatrix}.$$

9.1 Program Text

```
    Program f08qyfe

!     F08QYF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: f06uaf, nag_wp, x02ajf, ztrevc, ztrsna
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: eps, tnorm
      Integer                          :: i, info, ldt, ldvl, ldvr, ldwork, m, n
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: t(:,:), vl(:,:), vr(:,:), work(:,:)
      Real (Kind=nag_wp), Allocatable  :: rwork(:), s(:), sep(:)
      Logical                          :: select(1)
!     .. Executable Statements ..
      Write (nout,*) 'F08QYF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      ldt = n
      ldvl = n
      ldvr = n
      ldwork = n
      Allocate (t(ldt,n),vl(ldvl,n),vr(ldvr,n),work(ldwork,n+1),rwork(n),s(n), &
        sep(n))

!     Read T from data file

      Read (nin,*)(t(i,1:n),i=1,n)

!     Calculate the left and right eigenvectors of T

!     The NAG name equivalent of ztrevc is f08qxf
      Call ztrevc('Both','All',select,n,t,ldt,vl,ldvl,vr,ldvr,n,m,work,rwork, &
        info)

!     Estimate condition numbers for all the eigenvalues and right
!     eigenvectors of T

!     The NAG name equivalent of ztrsna is f08qyf
      Call ztrsna('Both','All',select,n,t,ldt,vl,ldvl,vr,ldvr,s,sep,n,m,work, &
        ldwork,rwork,info)

!     Print condition numbers of eigenvalues and right eigenvectors

      Write (nout,*) 'S'
      Write (nout,99999) s(1:m)
      Write (nout,*)
      Write (nout,*) 'SEP'
      Write (nout,99999) sep(1:m)

!     Calculate approximate error estimates (using the 1-norm)

      eps = x02ajf()
      tnorm = f06uaf('1-norm',n,n,t,ldt,rwork)
      Write (nout,*)
      Write (nout,*) 'Approximate error estimates for eigenvalues ', &
        'of T (machine-dependent)'
      Write (nout,99999)(eps*tnorm/s(i),i=1,m)
      Write (nout,*)
      Write (nout,*) 'Approximate error estimates for right ', &
```

```
        'eigenvectors of T (machine-dependent)'
      Write (nout,99999)(eps*tnorm/sep(i),i=1,m)

99999 Format ((3X,1P,7E11.1))
    End Program f08qyfe
```

9.2 Program Data

```
F08QYF Example Program Data
  4                                                              :Value of N
 (-6.0004,-6.9999) ( 0.3637,-0.3656) (-0.1880, 0.4787) ( 0.8785,-0.2539)
 ( 0.0000, 0.0000) (-5.0000, 2.0060) (-0.0307,-0.7217) (-0.2290, 0.1313)
 ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 7.9982,-0.9964) ( 0.9357, 0.5359)
 ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 3.0023,-3.9998)
                                                               :End of matrix T
```

9.3 Program Results

```
 F08QYF Example Program Results

 S
       9.9E-01    1.0E+00    9.8E-01    9.8E-01

 SEP
       8.4E+00    8.0E+00    5.8E+00    5.8E+00

 Approximate error estimates for eigenvalues of T (machine-dependent)
       1.0E-15    1.0E-15    1.1E-15    1.1E-15

 Approximate error estimates for right eigenvectors of T (machine-dependent)
       1.2E-16    1.3E-16    1.8E-16    1.8E-16
```